// File: donator.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.7.0 <0.9.0;

import "Ownable.sol";
import "IERC20.sol";

import "ICurve.sol";
import "IStakePrizePool.sol";

contract Donator is Ownable {
    event RecipientSet(address indexed recipient);
    event EtherReceived(address indexed sender, uint256 indexed value);
    event MinimumPriceSet(uint256 indexed price);
    event BaseTokenWithdrawn(uint256 indexed amount);
    event BaseTokenSwapped(uint256 indexed amount);
    event EtherDonated(address indexed recipient, uint256 indexed amount);

    /// @dev This offers an interface for the PoolTogether prize pool contract from which the prizes can be withdrawn
    IStakePrizePool public immutable stakePrizePool;

    /// @dev This is the prize token which is used by above prize pool
    IERC20 public immutable prizeToken;

    /// @dev This is the base token which is deposited and withdrawn from above prize pool
    IERC20 public immutable baseToken;

    /// @dev This is the pool used to swap the withdrawn prize token to ETH
    ICurve public immutable pool;

    /// @dev This defines the minimum price allowed when swapping the base token for ETH
    uint256 public minimumPrice;

    /// @dev This is the recipient of the donate function
    address public recipient;

    /// @dev Public constructor
    constructor (IStakePrizePool _stakePrizePool, IERC20 _prizeToken, IERC20 _baseToken, ICurve _pool, address _recipient) {
        require(address(_stakePrizePool) != address(0), "prize-pool-zero");
        require(address(_prizeToken) != address(0), "prize-token-zero");
        require(address(_baseToken) != address(0), "base-token-zero");
        require(address(_pool) != address(0), "pool-contract-zero");
        require(_recipient != address(0), "recipient-zero");

        stakePrizePool = _stakePrizePool;
        prizeToken = _prizeToken;
        baseToken = _baseToken;
        pool = _pool;

        recipient = _recipient;

        emit RecipientSet(recipient);
    }

    /// @notice Allows this contract to receive Ether
    receive() external payable {
        emit EtherReceived(msg.sender, msg.value);
    }

    /// @notice Allows the owner of this contract to set the receiving address
    /// @param _recipient The recipient for the donate function
    /// @dev Only allow setting a non-zero recipient and when the current balance is zero
    function setRecipient(address _recipient) external onlyOwner {
        require(_recipient != address(0), "recipient-zero");
        require(prizeToken.balanceOf(address(this)) == 0, "balance-not-zero");

        recipient = _recipient;

        emit RecipientSet(recipient);
    }

    /// @notice Allows the owner of this contract to set the minimum price for the swap (a.k.a. slippage)
    /// @param _price The minimum price
    function setMinimumPrice(uint256 _price) external onlyOwner {
        require(_price > 0, "price-zero");

        minimumPrice = _price;

        emit MinimumPriceSet(minimumPrice);
    }

    /// @notice Allows anyone to call the donate function
    function donate() external {
        require(minimumPrice > 0, "price-not-set");
        require(prizeToken.balanceOf(address(this)) > 0, "nothing-to-donate");

        uint256 baseTokenWithdraw = withdrawFromPool();
        emit BaseTokenWithdrawn(baseTokenWithdraw);

        uint256 etherSwapped = swapBaseTokenForEth(baseTokenWithdraw);
        emit BaseTokenSwapped(etherSwapped);

        // Send all ETH to the recipient
        (bool sent, ) = payable(recipient).call{value: etherSwapped}("");
        require(sent, "donate-failed");
        emit EtherDonated(recipient, etherSwapped);
    }

    /// @notice Helper function to withdraw the prize token from the pool
    function withdrawFromPool() internal returns (uint256 baseTokenWithdraw){
        stakePrizePool.withdrawInstantlyFrom(
            address(this),
            prizeToken.balanceOf(address(this)),
            address(prizeToken),
            0
        );
        baseTokenWithdraw = baseToken.balanceOf(address(this));
    }

    /// @notice Helper function to approve and swap the base token to ETH
    function swapBaseTokenForEth(uint256 _amountIn) internal returns (uint256 amountOut) {
        // Approve the withdrawn token for swapping
        baseToken.approve(address(pool), _amountIn);

        // Swap _amountIn of the token (index 1) for amountOutMinimum ETH (index 0)
        uint256 amountOutMinimum = _amountIn * minimumPrice / 1e18;
        amountOut = pool.exchange(1, 0, _amountIn, amountOutMinimum);
    }
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: ICurve.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.7.0 <0.9.0;

interface ICurve {
    function exchange(int128 i, int128 j, uint256 dx, uint256 minDy) external returns (uint256 dy);
}

// File: IStakePrizePool.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.7.0 <0.9.0;

interface IStakePrizePool {
    function withdrawInstantlyFrom(address from, uint256 amount, address controlledToken, uint256 maximumExitFee) external returns (uint256);
}